

Fearless programming and reasoning with infinities

Alexander Gryzlov

Research programmer
IMDEA Software Institute

Functional Programming Madrid

09/04/2024

Agenda

- Totality, partiality and fixed points
- Infinite data
- Interfaces and objects
- Automata
- The road ahead

Part 1

Totality, partiality and fixed points

Purely functional programming
= programming and reasoning with referentially
transparent higher-order functions

Having effects explicit simplifies reasoning

Purity

Very useful for reasoning about the intention and correctness

Types classify well-behaved programs, but we inevitably lose some programs

So there's a quest to regain expressivity by making type system more powerful

A crucial step is to unify types and programs

Totality

Pushing type-based reasoning further gives
dependent types

We can program the type-checker itself

```
1 data Vec (A :  $\mathcal{U}$ ) :  $\mathbb{N}$  →  $\mathcal{U}$  where
2   [] : Vec A zero
3   _::_ : A → Vec A n → Vec A (suc n)
4
5 data Format = Number Format
6             | Str Format
7             | Lit String Format
8             | End
9
10 PrintfType : Format →  $\mathcal{U}$ 
11 PrintfType (Number fmt) = (i : Int) → PrintfType fmt
12 PrintfType (Str fmt) = (str : String) → PrintfType fmt
13 PrintfType (Lit str fmt) = PrintfType fmt
14 PrintfType End = String
15
16 printfFmt : (fmt : Format) → (acc : String) → PrintfType fmt
17 printfFmt (Number fmt) acc = λ i → printfFmt fmt (acc ++ show i)
18 printfFmt (Str fmt) acc = λ str → printfFmt fmt (acc ++ str)
19 printfFmt (Lit lit fmt) acc = printfFmt fmt (acc ++ lit)
20 printfFmt End acc = acc
```

Totality

This however means that purity is not enough
Functions have to be total: defined everywhere

Otherwise the typechecking crashes or fails

Hanging terms = inconsistent

```
{-# TERMINATING #-}  
void : ⊥  
void = void  
  
oops : 2 + 2 = 5  
oops = absurd void
```

Semantics

- mathematical function: idealized pairing of input to outputs such that the output is uniquely determined by input (denotational)
- algorithmic function: a tree of instructions for manipulating abstract automata
- denotational view mostly ignores time, however operational view is typically quite low-level

Safety and liveness

Correctness properties typically split into

- safety (nothing bad ever happens)
- liveness (something good eventually happens)

Totality also has two aspects:

- defined input (no crashing)
- producing output (no hanging)

Safety reasoning

Partial functions violating safety = some arguments are not handled

Typically modelled with Maybe/Either

Violation of liveness = non-termination

The operational/temporal aspect (e.g. complexity) is generally hard to reason with in FP

Essentially, functions can drop and duplicate data in unrestricted fashion

Non-termination

Temporal aspects are typically "invisible"

How to model endless computation?

Total programming usually restricts to terminating functions

Too narrow, cannot reason about interactive programs

Need to model control flow in the type system

Non-termination hacks

We can try adding hacks:

1. construct individual terminating steps
2. make a small unsafe function that spins the steps
3. alternatively, add number of steps and then unsafely generate an infinite number

```
data Fuel = Dry | More Fuel

limit : ℕ → Fuel
limit zero = Dry
limit (suc n) = More (limit n)


{-# TERMINATING #-}
forever : Fuel
forever = More forever
```

Not very satisfactory, we should have a formal solution

Type-level time


A natural way of reasoning about time is to split it into steps/ticks on some global clock

The flow of time should be unidirectional



A thunk calculus

- Let us introduce a special type constructor \triangleright
- $\triangleright A$ is "A, but available one step later"
- Essentially a type-level thunk $() \Rightarrow A$
- Can also be thought of as staging
- The program generates a new program that runs after the first one and so on



Structure of later

```
next : A → ▷ A
ap   : ▷ (A → B) → ▷ A → ▷ B
```

It's an applicative functor
(will denote ap by \otimes)

Functorial structure

```
map : (A → B) → ▷ A → ▷ B
map f a▷ = next f ⊛ a▷
```


- We can derive a functorial action (will denote map by \bowtie)
- All the laws hold definitionally (by symbolic computation)

Not a monad

There is no monadic structure

`flatten : ▷ ▷ A → ▷ A`

This ensures that the temporal structure is preserved




For an arbitrary type there's also typically no

`▷ A → A`

Guarded recursion

- We can schedule computations, what now?
- Terminating → Productive
- Every recursive call is "guarded" by a thunk, giving back control
- Infinite / streaming computations, servers, OSs



Guarded recursion

```
fix: (▷ A → A) → A
```

- A form of Y-combinator
- Postulated definition, unfolding made propositional
- Can be safely erased down to the usual fixpoint

Ticked type theory

- We'll use Agda proof assistant for interactive examples
- Guarded modality is encoded as a function from `Ticks` of a modal type \mathbb{T}
- A proof of an elapsed time step
- Can encode `next`, `ap` and `map`

Ticked cubical type theory

- Technically we're also using the cubical mode of Agda
- No higher equalities / quotients / univalence
- Equality is encoded as a function from a continuous interval \mathbb{I}
- The interval is allowed to "time-travel"
- We can reason about the future in the present

Logical justification

- A flavour of provability modality
- 1933 - Gödel's analysis of S4
- 1950s - Löb's axiom: $\Box(\Box A \rightarrow A) \rightarrow \Box A$
- We're using the strong Löb's version: $(\Box A \rightarrow A) \rightarrow A$
- 1960s - GL system
- 1970s - intuitionistic systems, fixed points
- 2000s - links to formal semantics and category theory

Nakano's approximation

- Nakano, [2000] "A modality for recursion"
- Initially denoted ●
- Continued by many authors, most notably:
- Atkey-McBride'2013
- A series of papers by Birkedal and coauthors in 2010s
- Nowadays symbolized by a right triangle ▷
- Viewed as a special type constructor in a type system

Programming with \triangleright

So, to recap we have essentially 4 new constructs:

\triangleright , next, ap/\otimes , fix

(+ map/\otimes and some proof machinery)

What can we write?

Part 2

Infinite data

Which infinite types make sense?

Fitting the fix

Previously, I've said that for an arbitrary type there typically is no

▷ $A \rightarrow A$

However, that is the type of function we need:

```
fix: (▷ A → A) → A
```

We can construct such types with ▷

Partiality effect

- Recall the motivation of having a non-termination effect
- We can express it with two constructors + a guard
- Many names: L(ift), Event, De lay

```
data Part (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  now      : A → Part A
  later    : ▷ Part A → Part A
```



Partiality functor

```
map-body : (A → B)  
          → ▷ (Part A → Part B)  
          →   Part A → Part B
```

```
map-body f m▷ (now a) = now (f a)
```

```
map-body f m▷ (later p) = later (m▷ ⊗ p)
```

```
map : (A → B) → Part A → Part B
```

```
map f = fix (map-body f)
```

Mapping a function = waiting until the end and applying it

Partiality applicative

```
1 pure : A → Part A
2 pure = now
3
4 ap-body : ▷ (Part (A → B) → Part A → Part B)
5           → Part (A → B) → Part A → Part B
6 ap-body a▷ (now f)      (now x)      = now (f x)
7 ap-body a▷ (now f)      (later x▷)   = later (a▷ ⊗ next (now f) ⊗ x▷)
8 ap-body a▷ (later f▷)   (now x)      = later (a▷ ⊗ f▷ ⊗ next (now x))
9 ap-body a▷ (later f▷)   (later x▷)   = later (a▷ ⊗ f▷ ⊗ x▷)
10
11 ap : Part (A → B) → Part A → Part B
12 ap = fix ap-body
```

Unwind both structures "in parallel"

Partiality monad

```
flatten-body : ▷ (Part (Part A) → Part A)
              → Part (Part A) → Part A
flatten-body f▷ (now p)      = p
flatten-body f▷ (later p▷) = later (f▷ ⊗ p▷)

flatten : Part (Part A) → Part A
flatten = fix flatten-body
```

- Essentially an arbitrary sequence of nested ▷'s
- Reassociating → a monad
- $\triangleright \triangleright \triangleright (\triangleright \triangleright A) = \triangleright \triangleright \triangleright \triangleright \triangleright A$

Indexed partiality monad

Can be made more graphic by indexing with steps

```
mapd : (A → B) → Delayed A n → Delayed B n  
  
apd : Delayed (A → B) m  
      → Delayed A n  
      → Delayed B (max m n)
```

runs "in parallel"

```
_>>=d_ : Delayed A m  
          → (A → Delayed B n)  
          → Delayed B (m + n)
```

runs sequentially

Encoding applicative via bind changes complexity!

Partiality effect

```
1 never : Part ⊥
2 never = fix later
3
4 collatz-body : ▷ (ℕ → Part T) → ℕ → Part T
5 collatz-body c▷ 1 = now tt
6 collatz-body c▷ n =
7   if even n then later (c▷ ⊗ next (n ÷ 2))
8   else later (c▷ ⊗ next (suc (3 · n)))
9
10 collatz : ℕ → Part T
11 collatz = fix collatz-body
```

Wraps potentially non-terminating
computations

Conaturals

```
1 data N∞ : U where
2   ze : N∞
3   su : ▷ N∞ → N∞
4
5   infty : N∞
6   infty = fix su
7
8   +-body : ▷ (N∞ → N∞ → N∞) → N∞ → N∞ → N∞
9   +-body a▷      ze      ze      = ze
10  +-body a▷ x@(su _)   ze      = x
11  +-body a▷      ze      y@(su _) = y
12  +-body a▷      (su x▷)  (su y▷) =
13      su (next (su (a▷ ⊗ x▷ ⊗ y▷)))
14
15  _+_ : N∞ → N∞ → N∞
16  _+_ = fix +-body
```

Unary numbers extended with numerical infinity

≅ Part T

Conatural subtraction

```
÷-body : ▷ (ℕ∞ → ℕ∞ → Part ℕ∞) → ℕ∞ → ℕ∞ → Part ℕ∞
÷-body s▷ ze _ = now ze
÷-body s▷ x@(su _) ze = now x
÷-body s▷ (su x▷) (su y▷) = later (s▷ ⊗ x▷ ⊗ y▷)

_÷_ : ℕ∞ → ℕ∞ → Part ℕ∞
_÷_ = fix ÷-body

÷-infty : infty ÷c infty = never
...
```

(Saturating) subtraction is partial:

$\infty \div \infty$ never terminates

Co/free monad

Partiality is just an instantiation of the free monad with
the \triangleright functor

```
data Free (F :  $\mathcal{U} \rightarrow \mathcal{U}$ ) (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  Pure : A  $\rightarrow$  Free F A
  Roll : F (Free F A)  $\rightarrow$  Free F A


data Cofree (F :  $\mathcal{U} \rightarrow \mathcal{U}$ ) (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  Cof : A  $\rightarrow$  F (Cofree F A)  $\rightarrow$  Cofree F A
```

- Free monad is an F-branching tree with data on the leaves
- Cofree comonad is a tree with data at the branches
- What do we get by instantiating Cofree with \triangleright ?

Streams

```
data Stream (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where  
  cons : A  $\rightarrow$   $\triangleright$  Stream A  $\rightarrow$  Stream A
```

- Another classical infinite structure
- An inductive list with a delayed tail and no empty case
- A lazy linear producer of values



Stream functions

```
1 heads : Stream A → A
2 heads (cons x _) = x
3
4 tails : Stream A → Stream A
5 tails (cons _ xs) = xs
6
7 repeats : A → Stream A
8 repeats a = fix (cons a)
9
10 maps-body : (A → B)
11             → Stream A → Stream B
12             → Stream A → Stream B
13 maps-body f m as = cons (f (heads as)) (m ⊗ (tails as))
14
15 maps : (A → B) → Stream A → Stream B
16 maps f = fix (maps-body f)
17
18 natss-body : Stream ℕ → Stream ℕ
19 natss-body n = cons 0 (maps suc 0 n)
20
21 natss : Stream ℕ
22 natss = fix natss-body
```

Stream comonad


```
extracts : Stream A → A  
extracts = heads
```

```
duplicates-body : ▷ (Stream A → Stream (Stream A))  
                 → Stream A → Stream (Stream A)  
duplicates-body d▷ s@(cons _ t▷) = cons s (d▷ ⊗ t▷)
```

```
duplicates : Stream A → Stream (Stream A)  
duplicates = fix duplicates-body
```

extract = head

duplicate = tails



Causality

```
stutter : Stream A → Stream A
stutter = fix λ d▷ s →
  cons (heads s) (next (cons (heads s) (d▷ ⊗ tail▷s s)))

-- everyother : Stream A → Stream A
-- everyother = fix λ e▷ s →
--   cons (heads s) (e▷ ⊗ tail▷s (tail▷s s {!!}))
```



- We can define `stutter` but not `everyother`
- Violates causality
- (can be defined with clocks, however)

Folds and numbers

- We can define other familiar functions
- `foldr`, `scan`, `zipWith`, `interleave`
- numerical streams

```
1 fibs-body : ▷ Stream ℕ → Stream ℕ
2 fibs-body f▷ =
3   cons 0 ((λ s → cons 1 $ (zipWiths _+_ s) ∅ (tail▷s s)) ∅ f▷)
4
5 fibs : Stream ℕ
6 fibs = fix fibs-body
7
8 primess-body : ▷ Stream ℕ → Stream ℕ
9 primess-body p▷ = cons 2 ((maps suc ∘ scanl1s _·_) ∅ p▷)
10
11 primess : Stream ℕ
12 primess = fix primess-body
```

Part 3

Objects and interfaces

Let's look at the definition of the stream again

```
data Stream (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where  
  cons : A  $\rightarrow$   $\triangleright$  Stream A  $\rightarrow$  Stream A
```

A datatype with a single constructor is essentially
a record

Iterator

```
record Stream (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  constructor cons
  field
    hd   : A
    tl▷  : ▷ Stream A
```

We can treat the stream as an iterator object with two methods:


1. reading the head value
2. advancing by one step

Branching iterator

Can be generalized to an infinite binary tree

```
data Tree∞ (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  node : A → ▷ Tree∞ A → ▷ Tree∞ A
        → Tree∞ A

record Tree∞ (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  constructor node
  field
    val : A
    l▷ : ▷ Tree∞ A
    r▷ : ▷ Tree∞ A
```




Branching iterator

Or a rose tree with arbitrary branching

```
data RTree (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  rnode : A  $\rightarrow$  List ( $\triangleright$  RTree A)  $\rightarrow$  RTree A

record RTree (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  constructor rnode
  field
    val : A
    ch $\triangleright$  : List ( $\triangleright$  RTree A)
```



Terminating iterators

Multiple constructors makes this harder

```
1 data Colist (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
2   cnil    : Colist A
3   ccons  : A →  $\triangleright$  Colist A → Colist A
4
5 record Colist0 (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
6   constructor ccons0
7   field
8     hd    : Maybe A
9     tl $\triangleright$  :  $\triangleright$  Colist0 A
10
11 record Colist1 (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
12   constructor ccons1
13   field
14     hd    : A
15     emp? : Bool
16     tl $\triangleright$  :  $\triangleright$  Colist1 A
```

Set interface

How would we represent an infinite set of \mathbb{N} ?

(a finite set is usually some search structure `RedBlackTree` \mathbb{N})


Typically via a function $\mathbb{N} \rightarrow \text{Bool}$

We can also use `Stream Bool`

Generally, `Stream A` $\cong \mathbb{N} \rightarrow A$

Tabulation of a function

However, this is not very efficient



Set interface

Instead, we can encode a set interface as
a recursive guarded record

```
record SetN :  $\mathcal{U}$  where
  constructor mkSet
  field
    emp? : Bool
    has? :  $\mathbb{N} \rightarrow \text{Bool}$ 
    ins  :  $\mathbb{N} \rightarrow \triangleright \text{SetN}$ 
    uni  :  $\triangleright \text{SetN} \rightarrow \text{Part SetN}$ 
```

(actual implementation a bit more technical)

Strict positivity

- Guarded recursion has two general areas of application:
 1. Working with potentially infinite data structures
 2. Encoding non-strictly-positive recursive types
- **Strictly positive** type appears to the left of 0 arrows
- A syntactic approximation of monotonicity

Strict positivity

- **Strictly positive** type appears to the left of 0 arrows
- Another syntactic approximation
- **Positive** type appears in even positions
- **Negative** type appears in odd positions

```
data Expr :  $\mathcal{U} \rightarrow \mathcal{U}$  where
  Foo : ((Expr a  $\rightarrow$  Expr a)  $\rightarrow$  Expr b)  $\rightarrow$  Expr (a  $\rightarrow$  b)
          ^^^^^^----- positive occurrence
                ^^^^^^----- negative occurrence
                          ^^^^^^---- strictly positive occurrence
```


A finite set object

```
1 finiteSet-body : ▷ (List ℕ → Setℕ) → List ℕ → Setℕ
2 finiteSet-body f▷ l =
3   mkSet (empty? l)
4     (λ n → elem? n l)
5     (λ n → f▷ ⊗ next (n :: l))
6     (λ x▷ → later ((λ x →
7       foldrP (λ n z →
8         later (now ∅ (z .ins n))) x l) ∅ x▷))
9
10 finiteSet : List ℕ → Setℕ
11 finiteSet = fix finiteSet-body
```

Carries around the search structure (here a List)

An infinite set object

```
1 evensUnion-body : ▷ (SetN → SetN) → SetN → SetN
2 evensUnion-body e▷ s =
3   mkSet false
4     (λ n → even n or s .has? n)
5     (λ n → e▷ ⊗ s .ins n)
6     (λ x▷ → later ((λ f →
7       mapp f (s .uni x▷)) ∅ e▷))
8
9 evensUnion : SetN → SetN
10 evensUnion = fix evensUnion-body
```

Delegates to the parameter

Objects with IO

This idea can be extended to state and effects
Encode IO as a form of a partiality monad and
abstract over methods

```
1 record IOInt :  $\mathcal{U}$  (lsuc 0) where
2   field
3     Command :  $\mathcal{U}$ 
4     Response : Command  $\rightarrow$   $\mathcal{U}$ 
5
6 data IO (I : IOInt) (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
7   bnd : (c : Command I) (f : Response I c  $\rightarrow$   $\triangleright$  IO I A)  $\rightarrow$  IO I A
8   ret : (a : A)  $\rightarrow$  IO I A
9
10 record Interface :  $\mathcal{U}$  (lsuc 0) where
11   field
12     Method :  $\mathcal{U}$ 
13     Result : Method  $\rightarrow$   $\mathcal{U}$ 
14
15 record IOObj (Io : IOInt) (I : Interface) :  $\mathcal{U}$  where
16   field
17     mth : (m : Method I)  $\rightarrow$  IO Io (Result I m  $\times$  IOObj Io I)
```

Part 4

Automata

Let's go back to the idea of function tabulation

$$\text{Stream } A \cong \mathbb{N} \rightarrow A$$

What is the infinite tree isomorphic to?

```
data Tree∞ (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  node : A → ▷ Tree∞ A → ▷ Tree∞ A
        → Tree∞ A
```

Word consumers


$$\text{Tree}^\infty A \cong \mathbb{N}_2 \rightarrow A$$

(the type of binary numbers)

There's general construction to tabulate


$$T \rightarrow A \text{ into some } F A$$

where the structure of F mirrors that of T



Tries

We can think of structures as infinite tries whose branching factor is determined by T



Word automata

This idea can be generalized even further, to tabulated polymorphic functions:

```
data Stream (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  cons : A  $\rightarrow$  (T  $\rightarrow$   $\triangleright$  Stream A)  $\rightarrow$  Stream A

data Tree $^\infty$  (A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  cons : A  $\rightarrow$  (Bool  $\rightarrow$   $\triangleright$  Tree $^\infty$  A)  $\rightarrow$  Tree $^\infty$  A

data Moore (X A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  mre : A  $\rightarrow$  (X  $\rightarrow$   $\triangleright$  Moore X A)  $\rightarrow$  Moore X A
```

Word automata

```
data Moore (X A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where  
  mre : A  $\rightarrow$  (X  $\rightarrow$   $\triangleright$  Moore X A)  $\rightarrow$  Moore X A
```

$\text{Moore } X \ A \cong \text{List } X \rightarrow A$

Deterministic Moore automaton, common special case is

$\text{Moore } X \ \text{Bool} \cong \text{List } X \rightarrow \text{Bool}$

which is typically called a *recognizer*

Automata operations

```
1 pure : B → Moore A B
2 pure b = fix (pure-body b)
3
4 map : (B → C)
5       → Moore A B → Moore A C
6 ...
7
8 ap : Moore A (B → C) → Moore A B → Moore A C
9 ..
10
11 zipWith : (B → C → D)
12           → Moore A B → Moore A C → gMoore A D
13 zipWith f = ap ◦ map f
14
15 cat : Moore A B → Moore B C → Moore A C
16 ...
```

Regular expressions

```
1 Lang :  $\mathcal{U} \rightarrow \mathcal{U}$ 
2 Lang A = Moore A Bool
3
4  $\emptyset$  : Lang A
5  $\emptyset$  = pure false
6
7  $\varepsilon$  : Lang A
8  $\varepsilon$  = mre true  $\lambda \_ \rightarrow \emptyset$ 
9
10 char : A  $\rightarrow$  Lang A
11 char a = Mre false  $\lambda x \rightarrow$ 
12         if [ x  $\hat{=}$  a ] then  $\varepsilon$  else  $\emptyset$ 
13
14 compl : Lang A  $\rightarrow$  Lang A
15 compl = map not
16
17 _U_ : Lang A  $\rightarrow$  Lang A  $\rightarrow$  Lang A
18 _U_ = zipWith _or_
19
20 _ $\cap$ _ : Lang A  $\rightarrow$  Lang A  $\rightarrow$  Lang A
21 _ $\cap$ _ = zipWith _and_
```

Mealy automata

```
data Mealy (X A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  mly : (X → A × ▷ Mealy X A) → Mealy X A

data Moore (X A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  mre : A → (X → ▷ Moore X A) → Moore X A
```

Mealy X A \cong Stream X → Stream A
transducer automaton

Resumptions

```
data Res (I 0 A :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
  ret   : A → Res I 0 A
  cont  : (I → 0 × ▷ Res I 0 A) → Res I 0 A
```

- A Mealy automaton that possibly terminates
- a combination of a partiality and state monad

Coroutines

Passing control back and forth via thunks is conceptually programming with coroutines

Can be compiled into patterns of communication between consumer and producer automata

```
1 data Consume (A B :  $\mathcal{U}$ ) :  $\mathcal{U}$  where
2   end   : B → Consume A B
3   more : (A → ▷ Consume A B) → Consume A B
4
5 pipe-body : ▷ (Stream A → Consume A B → Part B)
6             → Stream A → Consume A B → Part B
7 pipe-body p▷ _ (end x) = now x
8 pipe-body p▷ (cons h t▷) (more f▷) = later (p▷ ⊗ t▷ ⊗ f▷ h)
9
10 pipe : Stream A → Consume A B → Part B
11 pipe = fix pipe-body
```

Part 5

The road ahead

Where to go next?

Clocked type theory

- We can work "under" thunks but not remove them
- Delayedness never decreases
- Weaker than proper coinduction
- Can be extended by a constant \square modality or clock variables to allow forcing "completed" thunks
- force : $(\forall \kappa \rightarrow \triangleright \kappa (A \kappa)) \rightarrow \forall \kappa \rightarrow A \kappa$
- Controlled violation of causality
- E.g. we can write every other function on streams

Bird's algorithm

- aka `replaceMin`
- later generalized to value recursion (`MonadFix`) in Haskell
- given a binary tree with data in leaves, replaces all values with a minimum in a single pass




Fig. 1




Fig. 2

Bird's algorithm

Classical form is somewhat weird

```
1 replaceMin :: Tree -> Tree
2
3 replaceMin t =
4   let (r, m) = rmb (t, m) in r
5   where
6     rmb :: (Tree, Int) -> (Tree, Int)
7     rmb (Leaf x, y) = (Leaf y, x)
8     rmb (Node l r, y) =
9       let (l', ml) = rmb (l, y)
10          (r', mr) = rmb (r, y)
11         in
12       (Node l' r', min ml mr)
```

Guarded decomposition

- We can decompose this in two temporal phases
- Compute the minimum and construct the thunk
- Then run the thunk
- Uses the feedback combinator
- $\text{feedback} : (\triangleright A \rightarrow B \times A) \rightarrow B$
 $\text{feedback } f = \text{fst } (\text{fix } (f \circ \triangleright \text{map } \text{snd}))$
- Inserts intermediate data between steps
- Cannot run the thunk without clocks

Continuations

- We can reason about control-flow based algorithms
- Harper's algorithm for matching on regexps with continuations
- Hofmann's algorithm for tree BFS

Hoffman's algorithm


- Typically done with queues
- Hoffman invented a purely functional continuation-based algorithm in 1993
- Requires an intermediate (non-strictly) positive datatype

```
data RouF (A :  $\mathcal{U}$ ) (R $\triangleright$  :  $\triangleright \mathcal{U}$ ) :  $\mathcal{U}$  where
  overRF : RouF A R $\triangleright$ 
  nextRF : (( $\triangleright$  R $\triangleright$   $\rightarrow$   $\triangleright$  Colist A)  $\rightarrow$  Colist A)  $\rightarrow$  RouF A R $\triangleright$ 


Rou :  $\mathcal{U} \rightarrow \mathcal{U}$ 
Rou A = fix (RouF A)
```

Breadth-first traversal

- Another form of a binary tree, data on both leaves and nodes
- Compute a breadth-first traversal



breadth-first traversal



depth-first traversal

Stream calculus

exact real numbers, series
stream differential equations

$$\frac{1}{1 - X} = 1 + X + X^2 + X^3 + \dots$$

Search algorithms

- sequential topology
- Tychonoff's theorem

```
→c-searchable' : (ds : is-discrete X) → searchable X
  → ((p , d) : d-predicate (Stream X))
  → (δ : ℕ) → δ is-u-mod-of p on (closenesss ds)
  → Σ[ s0 : Stream X ] (Σ (Stream X) p → p s0)
```

Vs coinduction

- Coinductive mechanisms are more liberal
- Productivity checker is purely syntactic
- Spend a few hours on a proof, get shot down
- Guarded constructions are type-directed

Conclusion

- A principled way to work with non-termination
- A common theme is overcoming syntactic approximations
- Thunks, streams, partiality
- Non-strictly positive datatypes
- Synthetic topology and domain theory
- Concurrency models (quotienting & cubical gizmos)

Working repos

- <https://github.com/clayrat/guarded-cm>
- <https://github.com/clayrat/guarded-termination>
- <https://github.com/clayrat/guarded-objects>
- <https://github.com/clayrat/guarded-automata>
- <https://github.com/clayrat/guarded-search>
- <https://github.com/clayrat/logrel-guarded>

Literature

- Nakano, [2000] "A modality for recursion"
- Artemov, Beklemishev, [2004] "Provability logic"
- <https://agda.readthedocs.io/en/latest/language/guarded.html>
- Atkey, McBride, [2013] "Productive Coprogramming with Guarded Recursion"
- Bird, [1984] "Using circular programs to eliminate multiple traversals of data"
- Berger, Matthes, Setzer, [2019] "Martin Hofmann's Case for Non-Strictly Positive Data Types"
- Clouston, Bizjak, Grathwohl, Birkedal, [2016] "The guarded lambda-calculus: Programming and reasoning with guarded recursion for coinductive types"
- Paviotti, Mogelberg, Birkedal, [2015] "A model of PCF in Guarded Type Theory"

Contacts

- <http://clayrat.github.io/>
- <https://software.imdea.org/~aliaksandr.hryzlou/>
- <https://www.linkedin.com/in/alexgryzlov/>
- <https://twitter.com/clayrat/>